

一次不定方程式（行列を利用して解を求める）

一次不定方程式 $ax+by=d$ の整数解を求める。

a を b で割った商を q , 余りを r とすると

$$\begin{pmatrix} a \\ b \end{pmatrix} = \begin{pmatrix} q & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} b \\ r \end{pmatrix}$$

$r \neq 0$ のとき, b を r で割った商を q_1 , 余りを r_1 とすると,

$$\begin{pmatrix} a \\ b \end{pmatrix} = \begin{pmatrix} q & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} q_1 & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} r \\ r_1 \end{pmatrix}$$

$r_1 \neq 0$ のとき, r を r_1 で割った商を q_2 , 余りを r_2 とすると,

$$\begin{pmatrix} a \\ b \end{pmatrix} = \begin{pmatrix} q & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} q_1 & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} q_2 & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} r_1 \\ r_2 \end{pmatrix}$$

以下, 同様にして

$r_k \neq 0$ のとき, r_{k-1} を r_k で割った商を q_{k+1} , 余りを r_{k+1} とすると,

$$\begin{pmatrix} a \\ b \end{pmatrix} = \begin{pmatrix} q & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} q_1 & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} q_2 & 1 \\ 1 & 0 \end{pmatrix} \dots \begin{pmatrix} q_{k+1} & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} r_k \\ r_{k+1} \end{pmatrix}$$

$r_n \neq 0$, $r_{n+1} = 0$ となるまで互除法の計算を続けると,

$$\begin{pmatrix} a \\ b \end{pmatrix} = \begin{pmatrix} q & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} q_1 & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} q_2 & 1 \\ 1 & 0 \end{pmatrix} \dots \begin{pmatrix} q_{n+1} & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} r_n \\ 0 \end{pmatrix}$$

ここで, $M = \begin{pmatrix} q & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} q_1 & 1 \\ 1 & 0 \end{pmatrix} \begin{pmatrix} q_2 & 1 \\ 1 & 0 \end{pmatrix} \dots \begin{pmatrix} q_{n+1} & 1 \\ 1 & 0 \end{pmatrix}$ とおいて

$$M^{-1} = \begin{pmatrix} x & y \\ u & v \end{pmatrix} \text{とすると,}$$

$$\begin{pmatrix} x & y \\ u & v \end{pmatrix} \begin{pmatrix} a \\ b \end{pmatrix} = \begin{pmatrix} d \\ 0 \end{pmatrix}$$

となるから,

$$ax+by=d$$

M と M^{-1} をプログラムで計算する。

Full BASIC の行列演算

DIM L(2,2),M(2,2), N(2,2) 2行2列の行列

MAT M=IDN M を単位行列で置き換える

MAT M=M*N M を積 MN で置き換える

MAT N=INV(M) N を M の逆行列 M^{-1} で置き換える

一つの MAT 文で一つの演算しか行えない。

互除法により，入力された 2 数の最大公約数を求めるプログラム

```
10 INPUT a,b
20 DO UNTIL b=0
30 LET r=MOD(a,b)
40 LET a=b
50 LET b=r
60 LOOP
70 PRINT a
80 END
```

a, b を入力して $ax+by=d$ となる x, y, d を求めるプログラム

```
100 DIM M(2,2),N(2,2)
110 INPUT a,b
120 LET aa=a
130 LET bb=b
140 MAT M=IDN
150 DO UNTIL b=0
160 LET q=INT(a/b)
170 LET r=MOD(a,b)
180 LET N(1,1)=q
190 LET N(1,2)=1
200 LET N(2,1)=1
210 LET N(2,2)=0
220 MAT M=M*N
230 LET a=b
240 LET b=r
250 LOOP
260 MAT N=INV(M)
270 PRINT aa;"*";N(1,1);"+";bb;"*";N(1,2);"="; a
280 END
```

280行を

```
280 PRINT N(1,1), N(1,2), b
```

にすれば120行と130行は不要。

問題 $M^{-1} = \begin{pmatrix} x & y \\ u & v \end{pmatrix}$ とおいたとき， x, y が整数になるのはなぜか？

ヒント q, q_1, q_2, \dots は整数なので， M の要素はすべて整数になる。